第二十六章　经济与金融中的优化问题

本章主要介绍用 LINGO 软件求解经济、金融和市场营销方面的几个优化问题的案例。

§1　经济均衡问题及其应用

在市场经济活动中，当市场上某种产品的价格越高时，生产商越是愿意扩大生产能力（供应能力），提供更多的产品满足市场需求；但市场价格太高时，消费者的消费欲望（需求能力）会下降。反之，当市场上某种商品的价格越低时，消费者的消费欲望（需求能力）会上升，但生产商的供应能力会下降。如果生产商的供应能力和消费者的需求能力长期不匹配，就会导致经济不稳定。在完全市场竞争的环境中，我们总是认为经济活动应当达到均衡（equilibrium），即生产和消费（供应能力和需求能力）达到平衡，不再发生变化，这时该商品的价格就是市场的清算价格。

下面考虑两个简单的单一市场及双边市场的具体实例，并介绍经济均衡思想在拍卖与投标问题、交通流分配问题中的应用案例。

1.1　单一生产商、单一消费者的情形

例 1　假设市场上只有一个生产商（记为甲）和一个消费者（记为乙）。对某种商品，他们在不同价格下的供应能力和需求能力如表 1 所示。举例来说，表中数据的含义是：当单价低于 2 万元但大于或等于 1 万元时，甲愿意生产 2t 产品，乙愿意购买 8t 产品；当单价等于或低于 9 万元但大于 4.5 万元时，乙愿意购买 2t 产品，甲愿意生产 8t 产品；依次类推。那么的市场价格应该是多少？

表 1　不同价格下的供应能力和需求能力

生产商（甲）		消费者（乙）	
单价（万元/t）	供应能力（t）	单价（万元/t）	需求能力（t）
1	2	9	2
2	4	4.5	4
3	6	3	6
4	8	2.25	8

（1）问题分析

仔细观察一下表 1 就可以看出来，这个具体问题的解是一目了然的：清算价格显然应该是 3 万元/t，因为此时供需平衡（都是 6t）。为了能够处理一般情况，下面通过建立优化模型来解决这个问题。

这个问题给人的第一印象似乎没有明确的目标函数，不太像是一个优化问题。不过，我们可以换一个角度来想问题：假设市场上还有一个虚拟的经销商，他是甲乙进行交易的中介。那么，为了使自己获得的利润最大，他将总是以可能的最低价格从甲购买产品，再以可能的最高价格卖给乙，直到进一步的交易无利可图为止。例如，最开始的

2t 产品他将会以 1 万元的单价从甲购买，以 9 万元的单价卖给乙；接下来的 2t 产品他会以 2 万元的单价从甲购买，再以 4.5 万元的单价卖给乙；再接下来的 2t 产品他只能以 3 万元的单价从甲购买，再以 3 万元的单价卖给乙（其实这次交易他已经只是保本，但我们仍然假设这笔交易会发生，例如他为了使自己的营业额尽量大）；最后，如果他继续购买甲的产品卖给乙，他一定会亏本，所以他肯定不会交易。因此，市场清算价格就是 3 万元。根据这个想法，我们就可以建立这个问题的线性规划模型。

（2）模型建立

决策变量：设甲以 1 万元，2 万元，3 万元，4 万元的单价售出的产品数量（单位：t）分别是 x_1, x_2, x_3, x_4，乙以 9 万元，4.5 万元，3 万元，2.25 万元的单价购买的产品数量（单位：t）分别是 y_1, y_2, y_3, y_4。

目标函数：就是虚拟经销商的总利润，即

$$9y_1 + 4.5y_2 + 3y_3 + 2.5y_4 - x_1 - 2x_2 - 3x_3 - 4x_4 \tag{1}$$

约束条件：

供需平衡 $$\sum_{i=1}^{4} x_i = \sum_{i=1}^{4} y_i \tag{2}$$

供应限制 $$x_i \le 2,\ i = 1,2,3,4 \tag{3}$$

消费限制 $$y_i \le 2,\ i = 1,2,3,4 \tag{4}$$

非负限制 $$x_i, y_i \ge 0,\ i = 1,2,3,4 \tag{5}$$

（3）模型求解

式（1）～（5）是一个线性规划模型，可以用 LINGO 求解，对应的 LINGO 程序如下：

```
model:
sets:
gx/1..4/:c1,c2,x,y;
endsets
data:
c1=1 2 3 4;
c2=9,4.5,3,2.5;
enddata
max=@sum(gx:c2*y-c1*x);
@sum(gx:x)=@sum(gx:y);
@for(gx:@bnd(0,x,2);@bnd(0,y,2));
```

```
end
```

求解这个模型，得到如下解答：

```
Global optimal solution found at iteration:           5
  Objective value:                              21.00000
                       Variable           Value        Reduced Cost
                         C1( 1)        1.000000            0.000000
                         C1( 2)        2.000000            0.000000
                         C1( 3)        3.000000            0.000000
                         C1( 4)        4.000000            0.000000
                         C2( 1)        9.000000            0.000000
                         C2( 2)        4.500000            0.000000
                         C2( 3)        3.000000            0.000000
                         C2( 4)        2.500000            0.000000
                          X( 1)        2.000000           -2.000000
                          X( 2)        2.000000           -1.000000
                          X( 3)        0.000000            0.000000
                          X( 4)        0.000000            1.000000
                          Y( 1)        2.000000           -6.000000
                          Y( 2)        2.000000           -1.500000
                          Y( 3)        0.000000            0.000000
                          Y( 4)        0.000000           0.5000000

                            Row    Slack or Surplus      Dual Price
                              1        21.00000            1.000000
                              2        0.000000           -3.000000
```

（4）结果解释

可以看出，最优解为 $x_1 = x_2 = y_1 = y_2 = 2$，$x_3 = x_4 = y_3 = y_4 = 0$。但你肯定觉得这还是没有解决问题，甚至认为这个模型错了，因为这个解法没有包括 3 万元单价的 2t 交易量。虽然容易验证 $x_1 = x_2 = x_3 = y_1 = y_2 = y_3 = 0$，$x_4 = y_4 = 0$ 也是最优解，但在一般情况下是难以保证一定求出这个解的。

那么如何才能确定清算价格呢？请仔细思考一下供需平衡约束（2）的对偶价格（dual prices）的含义。对偶价格又称影子价格，表示的是对应约束的右端项的价值。供需平衡约束目前的右端项为 0，影子价格为－3，意思就是说如果右端项增加一个很小的量（即甲的供应量增加一个很小的量），引起的经销商的损失就是这个小量的 3 倍。可见，此时的销售单价就是 3 万元，这就是清算价格。

（5）模型扩展

一般地，可以假设甲的供应能力随价格的变化情况分为 K 段，即价格位于区间

$[p_k, p_{k+1})$ 时，供应量最多为 c_k （ $k=1,2,\cdots,K$ ； $0<p_1<p_2<\cdots<p_{K+1}=\infty$ ； $0=c_0<c_1<c_2<\cdots<C_K$ ），我们把这个函数关系称为供应函数（这里它是一个阶梯函数）。同理，假设乙的消费能力随价格的变化情况分为 L 段，即价格位于区间 $(q_{j+1}, q_j]$ 时，消费量最多为 d_j（ $j=1,2,\cdots,L$；$q_1>\cdots>q_L>q_{L+1}=0$；$0=d_0<d_1<\cdots d_L$ ），我们把这个函数关系称为需求函数（这里它也是一个阶梯函数）。

设甲以 p_k 的价格售出的产品数量为 x_k（ $k=1,2,\cdots,K$ ），乙以 q_j 的价格购入的产品数量为 y_j（ $j=1,2,\cdots,L$ ）。记 $c_0=d_0=0$ ，则可以建立如下所示的线性规划模型：

$$\max \quad \sum_{j=1}^{L} q_j y_j - \sum_{k=1}^{K} p_k x_k \tag{6}$$

$$\text{s.t.} \quad \sum_{k=1}^{K} x_k - \sum_{j=1}^{L} y_j = 0 \tag{7}$$

$$0 \le x_k \le c_k - c_{k-1},\ k=1,2,\cdots,K \tag{8}$$

$$0 \le y_j \le d_j - d_{j-1},\ \ j=1,2,\cdots,L \tag{9}$$

1.2　两个生产商、两个消费者的情形

例 2　假设市场上除了例 1 中的甲和乙外，还有另一个生产商（记为丙）和另一个消费者（记为丁），他们在不同价格下的供应能力和需求能力如表 2 所示。此外，从甲销售到丁的每吨产品的运输成本是 1.5 万元，从丙销售到乙的每吨产品的运输成本是 2 万元，而甲、乙之间没有运输成本，丙、丁之间没有运输成本。这时，市场的清算价格应该是多少？甲和丙分别生产多少？乙和丁分别购买多少？

表 2　不同价格下的供应能力和消费能力

生产商（丙）		生产商（丁）	
单价（万元/t）	供应能力（t）	单价（万元/t）	供应能力（t）
2	1	15	1
4	4	8	3
6	8	5	6
8	12	3	10

（1）问题分析

首先，我们看看为什么要考虑从甲销售到丁的产品的运输成本和从丙销售到乙的产品的运输成本。如果不考虑这些运输成本，我们就可以认为甲乙丙丁处于同一个市场上，因此可以将两个生产商（甲和丙）的供应函数合并成一个供应函数，合并后就可以认为市场上仍然只有一个供应商。类似地，乙和丁的需求函数也可以合并成一个需求函数，合并后就可以认为市场上仍然只有一个消费者。这样，就回到了例 1 的情形。

也就是说，考虑运输成本在经济学上的含义，应当是认为甲乙是一个市场（地区或国家），而丙丁是另一个市场（地区或国家）。运输成本也可能还包括关税等成本，由于这个成本的存在，两个市场的清算价可能是不同的。

仍然按照例 1 的思路，可以建立这个问题的线性规划模型。

（2）模型的建立和求解

设甲以 1，2，3，4（万元）的单价售出的产品数量（单位：t）分别是 A_1, A_2, A_3, A_4，乙以 9，4.5，3，2.25（万元）的单价购买的产品数量（单位：t）分别是 X_1, X_2, X_3, X_4；丙以 2，4，6，8（万元）的单价售出的产品数量（单位：t）分别是 B_1, B_2, B_3, B_4，丁以 15，8，5，3（万元）的单价购买的产品数量（单位：t）分别是 Y_1, Y_2, Y_3, Y_4。此外，假设 AX 和 AY 分别是甲向乙和丁的供货量，BX 和 BY 分别是丙向乙和丁的供货量。这些决策变量之间的关系参见示意图 1。



图 1　决策变量之间的关系

目标函数仍然是虚拟经销商的总利润，约束条件仍然是四类(供需平衡、供应限制、需求限制和非负限制)，不过这时应注意供需平衡约束应该是包括图 1 所示的决策变量之间的关系：

$$AX + AY = A_1 + A_2 + A_3 + A_4 \tag{10}$$

$$BX + BY = B_1 + B_2 + B_3 + B_4 \tag{11}$$

$$AX + BX = X_1 + X_2 + X_3 + X_4 \tag{12}$$

$$AY + BY = Y_1 + Y_2 + Y_3 + Y_4 \tag{13}$$

此外的其它约束实际上只是一个简单的变量上界约束。

下面直接给出 LINGO 程序：

```
model:
sets:
num1/1..4/:c1,c2,c3,c4,d1,d2,A,B,X,Y;
num2/1,2/:AXY,BXY;
endsets
data:
c1=9 4.5 3 2.25;
c2=15 8 5 3;
c3=1 2 3 4;
c4=2 4 6 8;
d1=1 3 4 4;
d2=1 2 3 4;
enddata
max=@sum(num1:c1*x+c2*y-c3*A-c4*B)-2*BXY(1)-1.5*AXY(2);
@sum(num1:A)-@sum(num2:AXY)=0;
@sum(num1:B)-@sum(num2:BXY)=0;
AXY(1)+BXY(1)-@sum(num1:X)=0;
AXY(2)+BXY(2)-@sum(num1:Y)=0;
@for(num1:@bnd(0,A,2);@bnd(0,X,2);@bnd(0,B,d1);@bnd(0,Y,d2));
end
```

求解这个模型，得到如下解答：

```
 Global optimal solution found at iteration:           0
 Objective value:                                44.00000

                       Variable           Value        Reduced Cost
                        C1( 1)         9.000000            0.000000
                        C1( 2)         4.500000            0.000000
                        C1( 3)         3.000000            0.000000
                        C1( 4)         2.250000            0.000000
                        C2( 1)         15.00000            0.000000
                        C2( 2)         8.000000            0.000000
                        C2( 3)         5.000000            0.000000
                        C2( 4)         3.000000            0.000000
                        C3( 1)         1.000000            0.000000
                        C3( 2)         2.000000            0.000000
                        C3( 3)         3.000000            0.000000
```

```
         C3( 4)        4.000000            0.000000
         C4( 1)        2.000000            0.000000
         C4( 2)        4.000000            0.000000
         C4( 3)        6.000000            0.000000
         C4( 4)        8.000000            0.000000
         D1( 1)        1.000000            0.000000
         D1( 2)        3.000000            0.000000
         D1( 3)        4.000000            0.000000
         D1( 4)        4.000000            0.000000
         D2( 1)        1.000000            0.000000
         D2( 2)        2.000000            0.000000
         D2( 3)        3.000000            0.000000
         D2( 4)        4.000000            0.000000
          A( 1)        2.000000           -2.000000
          A( 2)        2.000000           -1.000000
          A( 3)        2.000000            0.000000
          A( 4)        0.000000            1.000000
          B( 1)        1.000000           -2.500000
          B( 2)        3.000000          -0.5000000
          B( 3)        0.000000            1.500000
          B( 4)        0.000000            3.500000
          X( 1)        2.000000           -6.000000
          X( 2)        2.000000           -1.500000
          X( 3)        0.000000            0.000000
          X( 4)        0.000000           0.7500000
          Y( 1)        1.000000           -10.50000
          Y( 2)        2.000000           -3.500000
          Y( 3)        3.000000          -0.5000000
          Y( 4)        0.000000            1.500000
        AXY( 1)        4.000000            0.000000
        AXY( 2)        2.000000            0.000000
        BXY( 1)        0.000000            3.500000
        BXY( 2)        4.000000            0.000000
            Row    Slack or Surplus      Dual Price
              1        44.00000            1.000000
              2        0.000000           -3.000000
              3        0.000000           -4.500000
              4        0.000000           -3.000000
              5        0.000000           -4.500000
```

（3）结果解释

可以看到，最优解为 $A_1 = A_2 = A_3 = X_1 = X_2 = 2$，$B_1 = 1$，$B_2 = 3$，$Y_1 = 1$，$Y_2 = 3$，$Y_3 = 3$，$AX = BY = 4$，$AY = 2$，$A_4 = B_3 = B_4 = X_3 = X_4 = Y_4 = BX = 0$。也就是说，甲将向丁销售2t产品，丙不会向乙销售。

那么如何才能确定清算价格呢？与例1类似，约束（2）是针对生产商甲的供需平衡条件，目前的右端项为0，影子价格为－3，意思就是说如果右端项增加一个很少的量（即甲的供应量增加一个很少的量），引起的经销商的损失就是这个小量的3.5倍。可见，此时甲的销售单价就是3万元，这就是甲面对的清算价格。

完全类似地，可以知道生产商丙面对的清算价格为4.5。自然地，乙面对的清算价格也是3，丁面对的清算价格也是4.5，因为甲乙位于同一个市场，而丙丁也位于同一个市场。这两个市场的清算价之差正好等于从甲、乙到丙、丁的运输成本（1.5），这是非常合理的。

（4）模型扩展

可以和1.1一样，将上面的具体模型一般化，即考虑供应函数和需求函数的分段数不是固定为4，而是任意有限正整数的情形。

很自然地，上面的方法很容易推广到不仅仅是2个市场，而是任意有限个市场的情形。理论上看这当然没有什么难度，只是这时变量会更多，数学表达式变得更复杂一些。

1.3　拍卖与投标问题

例3　假设一家拍卖行对委托的5类艺术品对外拍卖，采用在规定日期前投标人提交投标书的方式进行，最后收到了来自4个投标人的投标书。每类项目的数量、投标人对每个项目的投标价格如表3中所示。例如，有3件第4类艺术品；对每件第4类艺术品，投标人1，2，3愿意出的最高价分别为6，1，3，2（货币单位，如万元）。此外，假设每个投标人对每类艺术品最多只能购买1件，并且每个投标人购买的艺术品的总数不能超过3件。那么，哪些艺术品能够卖出去？卖给谁？这个拍卖和投标问题中每类物品的清算价应该是多少？

表3　拍卖与投标信息

招标项目类型		1	2	3	4	5
招标项目的数量		1	2	3	3	4
投标价格	投标人1	9	2	8	6	3
	投标人2	6	7	9	1	5
	投标人3	7	8	6	3	4
	投标人4	5	4	3	2	1

（1）问题分析

这个具体问题在实际中可能可以通过对所有投标的报价进行排序来解决，例如可以总是将艺术品优先卖给出价最高的投标人。但这种方法不太好确定每类艺术品的清算

价，所以我们这里还是借用前面两个例子中的方法，即假设有一个中间商希望最大化自己的利润，从而建立这个问题的线性规划模型。

（2）问题的一般提法和假设

先建立一般的模型，然后求解本例的具体问题，设有n类物品需要拍卖，第j类物品的数量为s_j（$j=1,2,\cdots,n$）；有m个投标者，投标者i（$i=1,2,\cdots m$）对第j类物品的投标价格为b_{ij}（假设非负）。投标者i对每类物品最多购买一件，且总件数不能超过c_i。我们的目标之一是要确定第j类物品的清算价格p_j，它应当满足下列假设条件：

i）成交的第j类物品的数量不超过s_j（$j=1,2,\cdots,n$）；

ii）对第j类物品的报价低于p_j的投标人将不能获得第j类物品；

iii）如果成交的第j类物品的数量少于s_j（$j=1,2,\cdots,n$）可以认为$p_j=0$（除非拍卖方另外指定一个最低的保护价）；

iv）对第j类物品的报价高于p_j的投标人有权获得第j类物品，但如果他有权获得的物品超过3件，那么我们假设他总是希望使自己的满意度最大（满意度可以用他的报价与市场清算价之差来衡量）。

（3）优化模型

用$0-1$变量x_{ij}表示是否分配一件第j类物品给投标者i，即$x_{ij}=1$表示分配，而$x_{ij}=0$表示不分配。目标函数仍然是虚拟的中间商的总利润（认为这些利润全部是拍卖行的利润也可以），即

$$\sum_{i=1}^{m}\sum_{j=1}^{n} b_{ij}x_{ij} \tag{14}$$

除变量取值为0或1的约束外，问题的约束条件主要是两类：每类物品的数量限制和每个投标人所能分到的物品的数量限制，即

$$\sum_{i=1}^{m} x_{ij} \le s_j,\ j=1,2,\cdots,n \tag{15}$$

$$\sum_{j=1}^{n} x_{ij} \le c_i,\ i=1,2,\cdots,m \tag{16}$$

模型就是在约束（15）、（16）下最大化目标函数（14）。

（4）模型求解

编写的LINGO程序如下：

```
MODEL:
TITLE 拍卖与投标;
SETS:
    AUCTION/1..5/: S;
    BIDDER/1..4/ : C;
    LINK(BIDDER,AUCTION): B, X;
ENDSETS
DATA:
    S=1 2 3 3 4;
    C=3 3 3 3;
    B=
```

9	2	8	6	3
6	7	9	1	5
7	8	6	3	4
5	4	3	2	1

```
;
ENDDATA
MAX=@SUM(LINK: B*X);
@FOR(AUCTION(J):
    [AUC_LIM] @SUM(BIDDER(I): X(I,J)) < S(J) );
@FOR(BIDDER(I):
    [BID_LIM] @SUM(AUCTION(J): X(I,J)) < C(I) );
@FOR(LINK: @BND(0,X,1));
END
```

需要指出的是，上面程序中DATA语句的数据表是直接从WORD文档中复制（Ctrl+C）后粘贴（Ctrl+V）过来的，所以显示格式继续保持了WORD文档的风格。

（5）求解结果解释

可以看到，最优解为：投标人1得到艺术品1，3，4，投标人2，3都得到艺术品2，3，5，投标人4得到艺术品4，5。结果，第4，5类艺术品各剩下1件没有成交。

那么如何才能确定清算价格呢？与例1和例2类似，约束“AUC_LIM”是针对每类艺术品的数量限制的，对应的影子价格就是其清算价格：即5类艺术品的清算价格分别是5，5，3，0，0。第4，5类艺术品有剩余，所以清算价格为0，这是符合前面的假设的。

可以指出的是：即使上面模型中不要求 x_{ij} 为 $0-1$ 变量（即只要求取0～1之间的实

数），由于这个问题的特殊性，最优解中 x_{ij} 也会要么取0，要么取1，不可能取0～1之间的其它数，所以可以将LINGO模型中“@BIN(X)”改为“@BND(0,X,1)”，这个线性规划的结果将与0－1整数线性规划得到的结果相同。

最后，大学生的选课问题与此是类似的，即把课程看成招标（拍卖）项目，而把学生愿意付出的选课费看成投标。据说国外有些大学的选课系统就是使用这个模型确定每门课程的清算价格（选课费用）的，而且取得了成功。

1.4　交通流均衡问题

例4　某地有如图2所示的一个公路网，每天上班时间有6千辆小汽车要从居民区 A 前往工作区 D。经过长期观察，我们得到了图中5条道路上每辆汽车的平均行驶时间和汽车流量之间的关系，如表4所示，那么，长期来看，这些汽车将如何在每条道路上分布？



图2　一个公路网示意图

表4　平均行驶时间和汽车流量之间的关系

道　路		AB	AC	BC	BD	CD
行驶时间（min）	流量≤2	20	52	12	52	20
	2＜流量≤3	30	53	13	53	30
	3＜流量≤4	40	54	14	54	40

（1）问题分析

这个问题看起来似乎与前面几个例子完全不同，但实际上交通流与市场经济活动类似，也存在着均衡。

我们可以想象有一个协调者，正如前面几个例子中的所谓中间商可以理解为市场规律一样，实际上这里的所谓协调者也可以认为是交通流的规律。交通流的规律就是每辆汽车都将选择使自己从 A 到 D 运行时间最少的路线，其必然的结果是无论走哪条路线从 A 到 D，最终花费的时间应该是一样的（否则，花费时间较长的那条线路上的部分汽车就会改变自己的路线，以缩短自己的行驶时间）。

也就是说，长期来看，这些汽车在每条道路上的分布将达到均衡状态（所谓均衡，这里的含义就是每辆汽车都不能仅仅通过自身独自改变道路，节省其行驶时间）。在这种想法下，我们来建立线性规划模型。

（2）优化模型

交通流的规律要求所有道路上的流量达到均衡，我们仍然类似例1和例2来考虑问题。如果车流量是一辆车一辆车增加的，那么在每条道路上车流量小于2时，车流量会

有一个分布规律；当某条道路上流量正好超过2时，新加入的一辆车需要选择使自己堵塞时间最短的道路。这就提示我们把同一条道路上的流量分布分解成不同性质的三个部分。也就是说，我们用$Y(AB)$表示道路AB上的总的流量，并进一步把它分解成三部分：

i）道路AB上的流量不超过2时的流量，用$X(2,AB)$表示；

ii）道路AB上的流量超过2但不超过3时，超过2的流量部分用$X(3,AB)$表示；

iii）道路AB上的流量超过3但不超过4时，超过3的流量部分用$X(4,AB)$表示。

依次类推，对道路AC,BC,BD,CD上同理可以定义类似的决策变量。因此，问题中总共有20个决策变量$Y(j)$和$X(i,j)$（$i=2,3,4$， $j=AB,AC,BC,BD,CD$）。

问题的目标应当是使总的堵塞时间最小。用$T(i,j)$表示流量$X(i,j)$对应的堵塞时间（即表3中的数据，是对每辆车而言的），我们看看用$\sum_{i=2,3,4}\sum_{j为道路}T(i,j)X(i,j)$作为总堵塞时间是否合适。很容易理解：后面加入道路的车辆可能又会造成前面进入道路的车辆的进一步堵塞，如流量为3时，原先流量为2的车辆实际上也只能按$T(3,j)$的时间通过，而不是$T(2,j)$。也就是说，$\sum_{i=2,3,4}\sum_{j为道路}T(i,j)X(i,j)$并不是总堵塞时间。但是我们也可以发现，$T(i,j)$关于$i$是单调增加的，即不断增加的车流只会使以前的堵塞加剧而不可能使以前的堵塞减缓。所以，关于决策变量$X(i,j)$而言，$\sum_{i=2,3,4}\sum_{j为道路}T(i,j)X(i,j)$与我们希望优化的目标的单调性是一致的。因此，可以用$\sum_{i=2,3,4}\sum_{j为道路}T(i,j)X(i,j)$作为目标函数进行优化。

约束条件有三类：

i）每条道路上的总流量Y等于该道路上的分流量X的和；

ii）道路交汇处A,B,C,D（一般称为节点）的流量守恒（即进入量等于流出量）；

iii）决策变量的上限限制，如$X(2,AB)\le 2$，$X(3,AB)\le 1$，$X(4,AB)\le 1$等。

于是对应的优化模型很容易直接写出（略）。

（3）模型求解

编写LINGO程序如下：

```
MODEL:
TITLE 交通流均衡;
SETS:
    ROAD/AB,AC,BC,BD,CD/:Y;
    CAR/2,3,4/;
    LINK(CAR,ROAD): T, X;
ENDSETS
DATA:
! 行驶时间(分钟);
    T=20,52,12,52,20
        30,53,13,53,30
      40,54,14,54,40;
ENDDATA
[OBJ] MIN=@SUM(LINK: T*X);  ! 目标函数;
! 四个节点的流量守恒条件;
[NODE_A] Y(@INDEX(AB))+Y(@INDEX(AC)) = 6;
[NODE_B] Y(@INDEX(AB))=Y(@INDEX(BC))+Y(@INDEX(BD));
[NODE_C] Y(@INDEX(AC))+Y(@INDEX(BC))=Y(@INDEX(CD));
[NODE_D] Y(@INDEX(BD))+Y(@INDEX(CD))=6;
! 每条道路上的总流量Y等于该道路上的分流量X的和;
@FOR( ROAD(I): [ROAD_LIM] @SUM(CAR(J): X(J,I)) = Y(I));
! 每条道路的分流量X的上下界设定;
@FOR(LINK(I,J)|I#EQ#1: @BND(0,X(I,J),2) );
@FOR(LINK(I,J)|I#GT#1: @BND(0,X(I,J),1) );
END
```

可以指出的是，上面4个节点的流量守恒条件中，其实只有3个是独立的（也就是说，第4个条件总可以从其它3个方程推导出来），因此从中去掉任何一个都不会影响到计算结果。

（4）结果解释

LINGO的运行结果表明，均衡时道路 AB, AC, BC, BD, CD 的流量分别是4，2，2，2，4（千辆）车。但是要注意，正如我们建立目标函数时所讨论过的，这时得到的目标函数值452并不是真正的总运行和堵塞时间，而是一个用来表示目标函数趋势的虚拟的量，没有太多实际物理意义。事实上，可以求出这时的真正运行时间是：每辆车通过 AB, AC, BC, BD, CD 道路分别需要40，52，12，52，40（min），也就是在图中三

条路线 $ABD, ACD, ABCD$ 上都需要92min，所以这也说明交通流确实达到了均衡。

于是，均衡时真正的总运行时间应该是 $6\times 92 = 552$ （千辆车·min）。

（5）模型讨论

仔细想想就会发现，上面的解并不是最优解，即均衡解并不一定是最优的流量分配方案。为了求出使所有汽车的总运行时间最小的交通流，应该如何做呢？也就是说，这相当于假设有一个权威的机构来统筹安排，最优地分配这些交通流，而不是像求均衡解时那样认为各个个体（每辆车）都可以自己选择道路，自然达到平衡状态。

为了进行统筹规划，我们需要把新增的流量 $X(i,j)$ （ $i = 2,3,4$ ， $j = AB, AC, BC, BD, CD$ ）造成的实际堵塞时间计算出来（仍按每辆车计算），而不是像上面那样不考虑对原有车流造成的堵塞效应。以道路 AB 为例。

i）当流量为2千辆时，每辆车的通过时间为20min，所以总通过时间是40（千辆车·min）；

ii）当流量增加一个单位（本题中一个单位就是1千辆）达到3千辆时，每辆车的通过时间为30min，所以总通过时间是90（千辆车·min）；

iii）当流量再增加一个单位达到4千辆时，每辆车的通过时间为40min，所以总通过时间是160（千辆车·min）。

由此可见，流量超过2而不超过3时，单位流量的增加导致的总通过时间的变化为90－40＝50（千辆车·min）；流量超过3而不超过4时，单位流量的增加导致的总通过时间的变化为160－90＝70（千辆车·min）。

类似地，对所有道路，都可以得到单位流量的增加导致总行驶时间的增量和汽车流量之间的关系（参加表5）。

表5　单位流量的增加导致总行驶时间的增量和汽车流量之间的关系

道　　路		AB	AC	BC	BD	CD
总行驶时间的增量（千辆车· min）	流量 ≤ 2	20	52	12	52	20
	$2 <$ 流量 ≤ 3	50	55	15	55	50
	$3 <$ 流量 ≤ 4	70	57	17	57	70

用表5中的总行驶时间的增量数据代替前面模型中的每辆车的行驶时间数据 $T(i,j)$，模型的其它部分完全不用变。重新求解LINGO模型，LINGO程序如下：

```
MODEL:
TITLE 交通流均衡;
SETS:
    ROAD/AB,AC,BC,BD,CD/:Y;
    CAR/2,3,4/;
```

```
    LINK(CAR,ROAD): T, X;
ENDSETS
DATA:
! 行驶时间(分钟);
    T=
```

20	52	12	52	20
50	55	15	55	50
70	57	17	57	70

```
;
ENDDATA
[OBJ] MIN=@SUM(LINK: T*X);  ! 目标函数;
! 四个节点的流量守恒条件;
[NODE_A] Y(@INDEX(AB))+Y(@INDEX(AC)) = 6;
[NODE_B] Y(@INDEX(AB))=Y(@INDEX(BC))+Y(@INDEX(BD));
[NODE_C] Y(@INDEX(AC))+Y(@INDEX(BC))=Y(@INDEX(CD));
[NODE_D] Y(@INDEX(BD))+Y(@INDEX(CD))=6;
! 每条道路上的总流量Y等于该道路上的分流量X的和;
@FOR( ROAD(I): [ROAD_LIM] @SUM(CAR(J): X(J,I)) = Y(I));
! 每条道路的分流量X的上下界设定;
@FOR(LINK(I,J)|I#EQ#1: @BND(0,X(I,J),2) );
@FOR(LINK(I,J)|I#GT#1: @BND(0,X(I,J),1) );
END
```

求得的最优车流分配方式是：道路 AB, AC, BD, CD 的流量都是3千辆，而道路 BC 上没有流量；总（加权）运行时间为498（千辆车·min），优于均衡时的结果552（千辆车·min）。此时，每辆车的运行时间＝498/6=83（min），少于均衡时的92min。当然，这个最优解必须强制执行，否则 AB 道路上的一些车到底 B 点时，发现当前走 BCD 的时间只需要 $12+30=42$（min），比走 BD 的时间（53min）短很多，所以他们就会改走 BCD，导致走 BCD 的时间（主要是走道路 CD 的时间）增加；如此下去，最后终将到达前面我们得到的均衡状态。

这是一个非常有趣的结果：当一个系统中的每个个体都独自追求个体利益最大化时，整体的利益却没有达到最大化。

更令人惊讶的是：这个例子的道路网中如果没有道路 BC，从 A 到 D 的平均时间是83min；而新开了一条道路 BC 以后，从 A 到 D 的平均时间居然变成92min，不是加快反而减慢了。由此也可以理解，做出一个科学、合理的交通网的规划是一件相当复杂的工作。

§2　投资组合问题

2.1　基本的投资组合模型

例5　美国某三种股票（A,B,C）12年（1943－1954）的价格（已经包括了分红在内）每年的增长情况如表6所示（表中还给出了相应年份的500种股票的价格指数的增长情况）。例如，表中第一个数据1.300的含义是股票A在1943年的年末价值是其年初价值的1.300倍，即收益为30%，其余数据的含义依此类推。假设你在1955年时有一笔资金准备投资这三种股票，并期望年收益率至少达到15%，那么你应当如何投资？当期望的年收益率变化时，投资组合和相应的风险如何变化？

表6　股票收益数据

年　份	股票A	股票B	股票C	股票指数
1943	1.300	1.225	1.149	1.258997
1944	1.103	1.290	1.260	1.197526
1945	1.216	1.216	1.419	1.364361
1946	0.954	0.728	0.922	0.919287
1947	0.929	1.144	1.169	1.057080
1948	1.056	1.107	0.965	1.055012
1949	1.038	1.321	1.133	1.187925
1950	1.089	1.305	1.732	1.317130
1951	1.090	1.195	1.021	1.240164
1952	1.083	1.390	1.131	1.183675
1953	1.035	0.928	1.006	0.990108
1954	1.176	1.715	1.908	1.526236

（1）问题分析

本例的问题称为投资组合（portfolio）问题，早在1952年Markowitz就给出了这个模型的基本框架，而且这个模型后来又得到了不断的研究和改进。一般来说，人们投资股票时的收益是不确定的，因此是一个随机变量，所以除了考虑收益的期望值外，还应当考虑风险。风险用什么衡量？Markowitz建议，风险可以用收益的方差（或标准差）来进行衡量：方差越大，则认为风险越大；方差越小，则认为风险越小。在一定的假设下，用收益的方差（或标准差）来衡量风险确实是合适的。为此，我们先对表6中给出的数据计算出三种股票收益的均值和方差（包括协方差）备用。

一种股票收益的均值衡量的是这种股票的平均收益状况，而收益的方差衡量的是这种股票收益的波动幅度，方差越大则波动越大（收益越不稳定）。两种股票收益的协方差表示的则是它们之间的相关程度：

i）协方差为0时两者不相关。

ii）协方差为正数表示两者正相关，协方差越大则正相关性越强（越有可能一赚皆赚，一赔俱赔）。

iii）协方差为负数表示两者负相关，绝对值越大则负相关性越强（越有可能一个赚，另一个赔）。

记股票 A,B,C 每年的收益率分别为 R_1,R_2,R_3（注意表中的数据减去1以后才是年收益率），则 R_i（$i=1,2,3$）是一个随机变量。用 E 和 D 分别表示随机变量的数学期望和方差算子，用cov表示两个随机变量的协方差（covariance），根据概率论的知识和表6给出的数据，则可以计算出年收益率的数学期望为

$$ER_1=0.0890833\text{，}ER_2=0.2136667\text{，}ER_3=0.2345833 \tag{17}$$

同样，可以计算股票 A,B,C 年收益率的协方差矩阵为

$$\mathrm{COV}=\begin{bmatrix} 0.01080754 & 0.01240721 & 0.01307513 \\ 0.01240721 & 0.05839170 & 0.05542639 \\ 0.01307513 & 0.05542639 & 0.09422681 \end{bmatrix} \tag{18}$$

计算的LINGO程序如下：

```
MODEL:
Title 均值向量Mean与协方差矩阵COV;
SETS:
  YEAR/1..12/;
  STOCKS/ A,  B,  C/: Mean;
  link(YEAR, STOCKS): R;
  temp/1..5/;
  tmatrix(YEAR,temp):tm;
  STST(Stocks,stocks): COV;
ENDSETS
DATA:
  tm =
   1943        1.300       1.225        1.149        1.258997
   1944        1.103       1.290        1.260        1.197526
   1945        1.216       1.216        1.419        1.364361
   1946        0.954       0.728        0.922        0.919287
   1947        0.929       1.144        1.169        1.057080
   1948        1.056       1.107        0.965        1.055012
   1949        1.038       1.321        1.133        1.187925
   1950        1.089       1.305        1.732        1.317130
   1951        1.090       1.195        1.021        1.240164
   1952        1.083       1.390        1.131        1.183675
   1953        1.035       0.928        1.006        0.990108
  1954        1.176       1.715        1.908        1.526236;
```

```
@text(data1.txt)=R;
@text(data2.txt)=Mean;
@text(data3.txt)=COV;
ENDDATA
CALC:  !计算均值向量Mean与协方差矩阵COV;
@for(tmatrix(i,j)|j #ge#2 #and# j #le# 4:R(i,j-1)=tm(i,j)-1);
@for(stocks(i): Mean(i) = @sum(year(j): R(j,i)) / @size(year) );
@for(stst(i,j): COV(i,j) = @sum(year(k):
(R(k,i)-mean(i))*(R(k,j)-mean(j))) / (@size(year)-1) );
ENDCALC
END
```

注意模型中计算协方差矩阵COV时，分母是样本数减去1，而不是样本数，这是常用的计算方法，主要是为了保持这个估计的无偏性(当然，样本数较大时两者差别不太)。

（2）模型建立

用决策变量 x_1, x_2 和 x_3 分别表示投资人投资股票 A, B, C 的比例。假设市场上没有其它投资渠道，且手上资金（可以不妨假设只有1个单位的资金）必须全部用于投资这三种股票，则

$$x_1, x_2, x_3 \geq 0\text{，}\ x_1 + x_2 + x_3 = 1 \tag{19}$$

年投资收益率 $R = x_1R_1 + x_2R_2 + x_3R_3$ 也是一个随机变量。根据概率论的知识，投资的总期望收益为

$$ER = x_1ER_1 + x_2ER_2 + x_3ER_3 \tag{20}$$

年投资收益率的方差为

$$\begin{aligned} V &= D(x_1R_1 + x_2R_2 + x_3R_3) \\ &= x_1^2DR_1 + x_2^2DR_2 + x_3^2DR_3 \\ &\quad + 2x_1x_2\operatorname{cov}(R_1, R_2) + 2x_1x_3\operatorname{cov}(R_1, R_3) + 2x_2x_3\operatorname{cov}(R_2, R_3) \\ &= \sum_{i=1}^{3}\sum_{j=1}^{3} x_ix_j\operatorname{cov}(R_i, R_j) \end{aligned} \tag{21}$$

实际的投资者可能面临许多约束条件，这里只考虑题中要求的年收益率（的数学期望）不低于15%，即

$$x_1ER_1 + x_2ER_2 + x_3ER_3 \geq 0.15 \tag{22}$$

所以，最后的优化模型就是在约束（19）和（22）下极小化（21）。由于目标函数 V 是决策变量的二次函数，而约束都是线性函数，所以这是一个二次规划问题。

（3）用LINGO求解模型

编写LINGO程序如下：

```
MODEL:
SETS:
   YEAR/1..12/;
   STOCKS/ A,  B,  C/: Mean,X;
   link(YEAR, STOCKS): R;
   STST(Stocks,stocks): COV;
ENDSETS
DATA:
TARGET=0.15;
Mean=@file(data2.txt);
COV=@file(data3.txt);
ENDDATA
[OBJ] MIN = @sum(STST(i,j): COV(i,j)*x(i)*x(j));
[ONE] @SUM(STOCKS: X) = 1;
[TWO] @SUM(stocks: mean*x) >= TARGET;
END
```

求得投资三种股票的比例大致是：A占53%，B占36%，C占11%。风险（方差）为0.02241378。

（4）用MATLAB软件对模型进行参数分析

对实际投资人来说，可能不仅希望知道指定的期望投资回报率下的风险（回报率的方差），可能更希望知道风险随着不同的投资回报率是如何变化的，然后作出最后的投资决策。这当然可以通过在上面的模型中不断修改约束中的参数（目前为0.15）来实现，如将0.15改为0.2345，则表示投资回报率希望达到23.45%（这几乎是可能达到的最大值了，因为这几乎是三种股票中最大的投资回报率，即股票C的回报率）。可以想到，这时应主要投资在股票C上。实际求解一下，可以知道最优解中投资股票C的份额大约是99.6%（剩余的大约0.4%投资在股票B上）。

目前LINGO软件还没有二次规划灵敏度分析的功能。下面我们利用MATLAB软件进行灵敏度分析，回报率的取值区间为[0.09,0.234]，变化步长为0.002。编写程序如下：

```
clc,clear
load data2.txt,load data3.txt
h=reshape(data3,[3,3]);
a=data2';
solution=[];
target=0.09;
hold on
while target<0.234
   [x,y]=quadprog(2*h,[],-a,-target,ones(1,3),1,zeros(3,1));
```

```
    plot(target,y,'*b');
    solution=[solution [target;x;y]];
    target=target+0.002;
    end
solution
```

得到的投资回报率与风险之间的关系曲线如图3所示



图3 投资回报率与风险之间的关系

从图3可以看出，投资回报率从22%附近开始，风险迅速增大。

2.2 存在无风险资产时的投资组合模型

例6 假设除了例5中的三种股票外，投资人还有一种无风险的投资方式，如购买国库券。假设国库券的年收益率为5%，如何考虑例5中的问题？

（1）问题分析

其实，无风险的投资方式（如国库券、银行存款等）是有风险的投资方式（如股票）的一种特例，所以这就意味着例5中的模型仍然是适用的。只不过无风险的投资方式的收益是固定的，所以方差（包括它与其它投资方式的收益的协方差）都是0。

（2）问题求解

假设国库券的投资方式记为 D，则当希望回报率为15%时，对应的LINGO模型如下：

```
MODEL:
Title 含有国库券的投资组合模型;
SETS:
  STOCKS1/ A,  B,  C/: mean1;
  STST1(Stocks1,stocks1): COV1;
  STOCKS/ A,  B,  C, D/: mean,X;
  STST(Stocks,stocks): COV;
ENDSETS
DATA:
mean=0.05;
```

```
COV=0;
COV1=@file(data3.txt);
Mean1=@file(data2.txt);
TARGET = 0.15; ! 0.10;
ENDDATA
calc:
@for(STOCKS(i)|i #ne# 4: mean(i)=mean1(i));
@for(STST(i,j)| i #ne# 4 #and# j #ne# 4:COV(i,j)=COV1(i,j));
endcalc
[OBJ] MIN = @sum(STST(i,j): COV(i,j)*x(i)*x(j));
[ONE] @SUM(STOCKS: X) = 1;
[TWO] @SUM(stocks: mean*x) >= TARGET;
END
```

计算结果为，投资 A 占8.7%， B 占42.9%， C 占14.3%， D （国库券）占34.1%，风险（方差）为0.02080347。与例5中的风险（方差为0.02241378）比较，无风险资产的存在可以使得投资风险减少。虽然国库券的收益率只有5%，比希望得到的收益率15%小很多，但在国库券上的投资要占到34.1%，其原因就是为了减少风险。

现在，我们把上面模型中的期望收益减少到10%，即把数据段中的语句“TARGET = 0.15”改为“TARGET = 0.10”，重新求解模型。计算结果如下：

投资 A 占4.3%， B 占21.4%， C 占7.2%， D （国库券）占67.1%，此时风险（方差为0.0052）进一步下降。请特别注意：你能发现这个结果（这里不妨称为结果2）与刚才“TARGET = 0.15”的结果（这里不妨称为结果1）有什么联系吗？

仔细观察这两个结果，可以发现：结果2中投资在有风险资产（股票 A,B,C ）上的比例大约都是结果1中相应的比例的一半。也就是说，无论你的期望收益和风险偏好如何，你手上所持有的风险资产本身相互之间的比例居然是不变的！变化的只是投资于风险资产与无风险资产之间的比例。有趣的是，这一现象在一般情况下也是成立的，一般称为“分离定理”，即风险资产之间的投资比例与期望收益和风险偏好无关。1981年诺贝尔经济学奖得主Tobin教授之所以获奖，很大一部分原因就是因为他发现了这个重要的规律。

也正是由于有这样一个重要结果，我们在下面各节的讨论中就不再考虑存在无风险资产的情形了，而只考虑确定风险资产之间的投资比例。

2.3 考虑交易成本的投资组合模型

例7 继续考虑例5（期望收益仍定为15%）。假设你目前持有的股票比例为：股票 A 占50%， B 占35%， C 占15%。这个比例与例5中得到的最优解有所不同。实际股票市场上每次股票买卖通常总有交易费，例如按交易额的1%收取交易费，这时你是否仍需要对所持的股票进行买卖（换手），以便满足最优解的要求？

（1）建立模型

仍用决策变量 x_1, x_2 和 x_3 分别表示投资人应当投资股票 A、B、C 的比例，进一步假设购买股票 A、B、C 的比例为 y_1, y_2 和 y_3，卖出股票 A、B、C 的比例为 z_1, z_2 和 z_3。其中，y_i 和 z_i（$i=1,2,3$）中显然最多只能有一个严格取正数，且

$$x_i, y_i, z_i \geq 0,\ i=1,2,3 \tag{23}$$

由于交易费用的存在，这时约束 $x_1+x_2+x_3=1$ 不一定还成立（只有不进行股票买卖，即 $y_1=y_2=y_3=z_1=z_2=z_3=0$ 时，这个约束才成立）。其实，这个关系式的本质是：当前持有的总资金是守恒的，在有交易成本（1%）的情况下，应当表示成如下形式：

$$\sum_{i=1}^{3} x_i + 0.01 \cdot \sum_{i=1}^{3} (y_i + z_i) = 1 \tag{24}$$

另外，考虑到当前持有的各只股票的份额 c_i, x_i, y_i 与 z_i（$i=1,2,3$）之间也应该满足守恒关系式

$$x_i = c_i + y_i - z_i,\ i=1,2,3 \tag{25}$$

这就是新问题的约束条件，模型的其它部分不用改变。

（2）模型求解

问题对应的LINGO程序如下：

```
MODEL:
Title 考虑交易费的投资组合模型;
SETS:
   STOCKS/ A,  B,  C/: C,Mean,X,Y,Z;
   STST(Stocks,stocks): COV;
ENDSETS
DATA:
! 股票的初始份额;
c=0.5  0.35  0.15;
TARGET=0.15;
Mean=@file(data2.txt);
COV=@file(data3.txt);
ENDDATA
[OBJ] MIN = @sum(STST(i,j): COV(i,j)*x(i)*x(j));
[ONE] @SUM(STOCKS: X+0.01*Y+0.01*Z) = 1;
```

```
[TWO] @SUM(stocks: mean*x) >= TARGET;
@FOR(stocks: [ADD] x = c + y - z);
END
```

在这个LINGO模型中，股票C是原始集合“STOCKS”的一个元素，不会因为与集合的属性C同名而混淆。这是LINGO新版本比LINGO旧版本的一个改进之处。

2.4 利用股票指数简化投资组合模型

例8 继续考虑例5（期望收益率仍定为15%）。在实际的股票市场上，一般存在成千上万的股票，这时计算两两之间的相关性（协方差矩阵）将是一件非常费事甚至不可能的事情。例如，1000只股票就需要计算 $C_{1000}^2 = 499500$ 个协方差。能否通过一定方式避免协方差的计算，对模型进行简化呢？例如，例5中还给出了当时股票指数的信息，但我们到此为止一直没有利用。我们这一节就考虑利用股票指数对前面的模型进行修改和简化。

（1）问题分析

可以认为股票指数反映的是股票市场的大势信息，对具体每只股票的涨跌通常是有显著影响的。我们这里最简单地假设每只股票的收益与股票指数成线性关系，从而可以通过线性回归方法找出这个线性关系。

（2）线性回归

具体地说，用 M 表示股票指数（也是一个随机变量），其均值为 $m_0 = E(M)$，方差为 $s_0^2 = D(M)$。根据上面的线性关系的假定，对某只具体的股票 i，其价值 R_i（随机变量）可以表示成

$$R_i = u_i + b_i M + e_i \tag{26}$$

其中 u_i 和 b_i 需要根据所给数据经过回归计算得到，e_i 是一个随机误差项，其均值为 $E(e_i) = 0$，方差为 $s_i^2 = D(e_i)$。此外，假设随机误差项 e_i 与其它股票 j（$j \neq i$）和股票指数 M 都是独立的，所以 $E(e_i e_j) = E(e_i M) = 0$。

先看看如何根据所给数据经过回归计算得到 u_i 和 b_i。记所给的12年的数据为 $\{M^{(k)}, R_i^{(k)}\}$，（$k = 1,2,\cdots,12$），线性回归实际上是要使误差的平方和最小，即要解如下优化问题：

$$\min \sum_{k=1}^{12} (e_i^{(k)})^2 = \sum_{k=1}^{12} | u_i + b_i M^{(k)} - R_i^{(k)} |^2 \text{ , } i = 1,2,3 \tag{27}$$

对这里给出的三种股票，可以编写如下LINGO程序求出线性回归的系数 u_i 和 b_i（同时也在计算（CALC）段计算 M 的均值 m_0 和方差 s_0^2，标准差 s_0 的值）：

```
MODEL:
Title 线性回归模型;
SETS:
   YEAR/1..12/:M;
   STOCKS/A,  B,  C/: u, b, s2, s;
   temp/1..5/;
   tmatrix(YEAR,temp):tm;
   link(YEAR, STOCKS): R, e;
ENDSETS
DATA:
   num=?;
   tm =
    1943         1.300        1.225         1.149          1.258997
    1944         1.103        1.290         1.260          1.197526
    1945         1.216        1.216         1.419          1.364361
    1946         0.954        0.728         0.922          0.919287
    1947         0.929        1.144         1.169          1.057080
    1948         1.056        1.107         0.965          1.055012
    1949         1.038        1.321         1.133          1.187925
    1950         1.089        1.305         1.732          1.317130
    1951         1.090        1.195         1.021          1.240164
    1952         1.083        1.390         1.131          1.183675
    1953         1.035        0.928         1.006          0.990108
   1954         1.176        1.715         1.908          1.526236;
ENDDATA
CALC:
@for(tmatrix(i,j)|j #ge#2 #and# j #le# 4:R(i,j-1)=tm(i,j));
@for(tmatrix(i,j)| j #eq# 5:M(i)=tm(i,j));
mean0=@sum(year: M)/@size(year);
s20=@sum(year: @sqr(M-mean0)) / (@size(year)-1);
s0=@sqrt(s20);
ENDCALC
```

```
[OBJ] MIN = @sum(stocks(i)|i#eq#num: s2(i));
@for(link(k,i)|i#eq#num: [ERROR] e(k,i) = R(k,i)-u(i)-b(i)*M(k));
@for(stocks(i)|i#eq#num:[VAR] s2(i)=(@sum(year(k): @sqr(e(k,i))) /
(@size(year)-2));
[STD] s(i)=@sqrt(s2(i)) );
@for(stocks: @free(u);@free(b) );
@for(link: @free(e) );
END
```

对上面的这个程序，请注意以下几点：

i）在CALC段直接计算了 M 的均值 m_0 和方差 s_0^2（为了使这个估计是无偏估计，分母是11而不是12）以及标准差 s_0。

ii）程序中使用了两个常用的数学函数：平方函数@sqr和平方根函数@sqrt。

iii）除了计算回归系数外，我们同时估计了回归误差的方差 s_i^2 和标准差 s_i。为了使这个估计是无偏估计，计算 s_i^2 时分母是10而不是11或12，这时因为此时估计了两个参数，自由度少了两个。

iv）@free(u)，@free(b)，@free(e)三个语句一定不能少，因为这几个变量不一定是非负的。

v）DATA段定义了一个变量num，并用“num=?”语句表示其具体值需要由使用者在程序运行时输入。变量num的作用是控制当前对哪只股票进行线性回归（num＝1，2，3分别对应于股票 A,B,C）。

vi）其实，这个问题也可以对三只股票的回归不加区分，即放在同一个模型中同时优化（相应地，只需要去掉上面程序中的控制变量num和所有的过滤条件“i#eq#num”），不过这样就会增加变量的个数，我们不建议大家那样做。也就是说，对于能够分解成小规模问题的优化问题，最好一个一个分开做，这样可以减少问题规模，有助于求到比较好的解。

运行上述LINGO程序，得到的计算结果为：股票指数 M 的均值 $m_0 = 1.191458$，方差为 $s_0^2 = 0.02873661$，标准差为 $s_0 = 0.1695188$；对股票 A，回归系数 $u_1 = 0.5639761$，$b_1 = 0.4407264$，误差的方差 $s_1^2 = 0.00574832$，误差的标准差 $s_1 = 0.07581767$。

同理（运行时输入num=2或3），可以得到：对股票 B，回归系数 $u_2 = -0.2635028$，$b_2 = 1.239799$，误差的方差 $s_2^2 = 0.01564263$，误差的标准差 $s_2 = 0.1250705$。对股

票C，回归系数$u_3=-0.5809592$，$b_3=1.523798$，误差的方差$s_3^2=0.03025165$，误差的标准差$s_3=0.17393$。

（3）优化模型

现在，仍用决策变量x_1,x_2和x_3分别表示投资人应当投资股票A,B,C的比例，其中

$$x_1,x_2,x_3\ge 0，\ x_1+x_2+x_3=1 \tag{28}$$

此时，与2.1节的讨论类似，对应的收益应该表示为

$$R=\sum_{i=1}^{3}x_iR_i=\sum_{i=1}^{3}x_i(u_i+b_iM+e_i) \tag{29}$$

收益的期望为

$$ER=\sum_{i=1}^{3}x_iE(u_i+b_iM+e_i)=\sum_{i=1}^{3}x_i(u_i+b_im_0) \tag{30}$$

收益的方差为

$$DR=\sum_{i=1}^{3}x_i^2D(u_i+b_iM+e_i)=\sum_{i=1}^{3}(b_i^2s_0^2+s_i^2)x_i^2 \tag{31}$$

综上所述，建立如下模型

$$\min\sum_{i=1}^{3}(b_i^2s_0^2+s_i^2)x_i^2 \tag{32}$$

$$\text{s.t.}\ \ \sum_{i=1}^{3}x_i=1， \tag{33}$$

$$\sum_{i=1}^{3}(u_i+b_im_0)x_i\ge 0.15 \tag{34}$$

$$x_i\ge 0，\ i=1,2,3 \tag{35}$$

（4）模型求解

为了数据传递方便，我们把三个回归模型同时计算。LINGO程序如下：

```
MODEL:
Title 线性回归模型;
SETS:
```

```
   YEAR/1..12/:M;
   STOCKS/A,  B,  C/: u, b, s2, s;
   temp/1..5/;
   tmatrix(YEAR,temp):tm;
   link(YEAR, STOCKS): R, e;
ENDSETS
DATA:
tm =
    1943         1.300        1.225         1.149          1.258997
    1944         1.103        1.290         1.260          1.197526
    1945         1.216        1.216         1.419          1.364361
    1946         0.954        0.728         0.922          0.919287
    1947         0.929        1.144         1.169          1.057080
    1948         1.056        1.107         0.965          1.055012
    1949         1.038        1.321         1.133          1.187925
    1950         1.089        1.305         1.732          1.317130
    1951         1.090        1.195         1.021          1.240164
    1952         1.083        1.390         1.131          1.183675
    1953         1.035        0.928         1.006          0.990108
   1954         1.176        1.715         1.908          1.526236;
@text('data4.txt')=s2;
@text('data5.txt')=u;
@text('data6.txt')=b;
ENDDATA
CALC:
@for(tmatrix(i,j)|j #ge#2 #and# j #le# 4:R(i,j-1)=tm(i,j));
@for(tmatrix(i,j)| j #eq# 5:M(i)=tm(i,j));
mean0=@sum(year: M)/@size(year);
s20=@sum(year: @sqr(M-mean0)) / (@size(year)-1);
s0=@sqrt(s20);
ENDCALC
[OBJ] MIN = @sum(stocks(i): s2(i));
@for(link(k,i): [ERROR] e(k,i) = R(k,i)-u(i)-b(i)*M(k));
@for(stocks(i):[VAR] s2(i)=(@sum(year(k): @sqr(e(k,i))) / (@size(year)-2));
[STD] s(i)=@sqrt(s2(i)) );
@for(stocks: @free(u);@free(b) );
@for(link: @free(e) );
END
```

二次规划（32）～（35）的LINGO程序如下：

```
MODEL:
Title 利用股票指数简化投资组合模型;
SETS:
   STOCKS/A,  B,  C/: u, b, s2, x;
ENDSETS
DATA:
mean0=1.191458;
s20 = 0.02873661;
s2  =  @file(data4.txt);
u   =  @file(data5.txt);
b   =  @file(data6.txt);
ENDDATA
[OBJ] MIN = @sum(stocks: (@sqr(b)*s20+s2)*@sqr(x));
@sum(stocks: x)=1;
@sum(stocks: (u+b*mean0)*x)>1.15;
END
```

计算结果为，最后的持股情况是：A 大约占初始时刻总资产的54%，B 占27%，C 占19%。这个结果与例5的结果是不同的。

2.5 其它目标下的投资组合模型

前面介绍的模型中都是在可能获得的收益的数学期望满足一定最低要求的前提下，用可能获得的收益的方差来衡量投资风险，将其作为最小化的目标。这种做法的合理性通常至少需要有两个基本假设：

（1）可能获得的收益的分布是对称的（如正态分布）。因为这时未来收益高于设定的最低要求和低于设定的最低要求的数量和概率是一样的。可惜的是，实际中这个假设往往难以验证。

（2）投资者对风险（或偏好）的效用函数是二次的，否则为什么值选择效益（随机变量）的二阶矩（方差）来衡量风险使之最小化，而不采用其它阶数的矩？

一般来说，投资者实际关心的通常是未来收益低于设定的最低要求的数量（即低多少）和概率，也就是说更关心的是下侧风险（downside risk）。所以，如果分布不是对称的，则采用收益的方差来衡量投资风险就不一定合适。为了克服这个缺陷，可以用收益低于最低要求的数量的均值（一阶矩）作为下侧风险的衡量依据，即作为最小化的目标。此外，也可以采用收益低于最低要求的数量的二阶矩（即收益的半方差，semivariance）作为衡量投资风险的依据。其实，半方差计算与方差计算类似，只是只有当收益低于最低要求的收益率时，才把两者之差的平方记入总风险，而对收益高于最低要求的收益率时的数据忽略不计。这方面的具体模型这里就不再详细介绍了。

下面介绍一个与上面这些优化目标完全不同的投资组合模型，这个模型虽然很简单，但却会产生一些非常有趣的现象。

例9　假设市场上只有两只股票 A、B 可供某个投资者购买，且该投资者对未来一

年的股票市场进行了仔细分析，认为市场只能出现两种可能的情况（1和2）。此外，该投资者对每种情况出现的概率、每种情况出现时两只股票的增值情况都进行了预测和分析（见表7，可以看出股票 A、B 的均值和方差都是一样的）。该投资者是一位非常保守的投资人，其投资目标是使两种情况下最小的收益最大化（也就是说，不管未来发生哪种情况，他都能至少获得这个收益）。如何建立模型和求解？

表7　两种情况出现的概率及两只股票的增值情况

情形	发生概率	股票 A	股票 B
1	0.8	1.0	1.2
2	0.2	1.5	0.7

（1）优化模型与求解

设年初投资股票 A、B 的比例分别为 x_1, x_2，决策变量 x_1, x_2 显然应该满足

$$x_1, x_2 \geq 0, \quad x_1 + x_2 = 1 \tag{36}$$

此外，使最小收益最大的“保守”目标实际上就是希望：

$$\max\{\min(1.0x_1 + 1.2x_2, 1.5x_1 + 0.7x_2)\} \tag{37}$$

引入一个辅助变量 $y = \min(1.0x_1 + 1.2x_2, 1.5x_1 + 0.7x_2)$，这个模型就可以线性化为

$$\max\ y$$

$$\text{s.t.}\quad x_1 + x_2 = 1$$

$$x_1 + 1.2x_2 \geq y$$

$$1.5x_1 + 0.7x_2 \geq y$$

编写LINGO程序如下：

```
model:
sets:
COL/1..2/:x;
ROW/1..2/;
link(ROW,COL):a;
endsets
data:
a=1 1.2 1.5 0.7;
enddata
max=y;
```

```
@sum(COL:x)=1;
@for(ROW(i):@sum(COL(j):a(i,j)*x(j))>y);
End
```

可见，此时应该投资 A、B 股票各50%，至少可以增值10%。

（2）讨论

现在，假设有一位绝对可靠的朋友告诉该投资者一条重要信息：如果情形1发生，股票 B 的增值将达到30%而不是表7中给出的20%。那么，一般人的想法应该是增加对股票 B 的持有份额。果真如此吗？这个投资人如果将上面模型中的1.2改为1.3计算，将得到如下结果：$x_1 = 0.5454545$，$x_2 = 0.4545455$，$y = 1.136364$。

也就是说，应该减少对股票 B 的持有份额，增加对股票 A 的持有份额。这真是叫人大吃一惊！这相当于说：有人告诉你有某只股票涨幅要增加了，你赶紧说：那我马上把这只股票再卖点吧。之所以出现如此奇怪的现象，就是由于这个例子中的目标的特殊性引起的：我们可以看到新的解可以保证增值达到13.6364%，确实比原来的10%增加了。

最后需要指出：我们上面所有关于投资组合的这些讨论基本上只是纯技术面的讨论，只利用历史数据来说话，认为历史数据中包含了引起股票涨跌的所有因素。在实际股票市场上，影响股票涨跌的因素可能有很多（如政策变化、银行加息、能源短缺、技术进步等），未来不长时间内可能发生的一些重大事件很可能以前没有发生过，因此也不可能体现在历史数据中。所以，进行投资选择前，还应该进行基本面分析，需要对未来的一些重要影响因素、重大事件发生的可能性及其对每种股票涨跌的影响进行预测和分析，最后综合利用历史数据和这些预测数据，决定投资组合。如何将这些预测数据与历史数据一起使用，建立相应的投资组合模型，这里就不再更多地介绍了。这方面的模型有很多，有兴趣的可以继续查阅相关的专业书籍和研究文献。

§3　市场营销问题

3.1　新产品的市场预测

例10　某公司开发了一种新产品，打算与目前市场上已有的三种同类产品竞争。为了了解这种新产品在市场上的竞争力，在大规模投放市场前，公司营销部门进行了广泛的市场调查，得到了表8。四种产品分别记为 A、B、C、D，其中 A 为新产品，表中的数据的含义是：最近购买某种产品（用行表示）的顾客下次购买四种产品的机会（概率）。例如：表中第一行数据表示当前购买产品 A 的顾客，下次购买产品 A、B、C、D 的概率分别为75%，10%，5%，10%。请你根据这个调查结果，分析新产品 A 未来的市场份额大概是多少？

表8　市场调查数据

产品	A	B	C	D
A	0.75	0.1	0.05	0.1
B	0.4	0.2	0.1	0.3
C	0.1	0.2	0.4	0.3

D	0.2	0.2	0.3	0.3

（1）问题分析

新产品进入市场后，初期的市场份额将会不断发生变化，因此，本例中的问题是一个离散动态随机过程，也就是马氏链（Markov chain）。很显然，上面给出的表实际上是转移概率矩阵（注意每行元素的和肯定为1）。要分析新产品 A 未来的市场份额，就是要计算稳定状态下每种产品的概率。

（2）模型的建立

记 N 为产品种数。产品编号为 i（$i=1,2,\cdots,N$），转移概率矩阵的元素记为 T_{ij}，稳定状态下产品 i 的市场份额记为 p_i。

因为是稳定状态，所以应该有

$$p_i=\sum_{j=1}^{n}p_jT_{ji}\text{，}\quad i=1,2,\cdots,N \tag{38}$$

不过，这 N 个方程实际上并不独立，至少有一个是冗余的。好在我们还有另一个约束，即 N 种产品的市场份额之和等于1

$$\sum_{j=1}^{N}p_j=1 \tag{39}$$

可见，这个问题的模型实际上是一个非常简单的方程组（当然，还应该增加概率 p_i 非负的约束）。如果把这些看成约束条件，那就是一个特殊的优化模型（没有目标函数）。

（3）模型的求解

LINGO程序如下：

```
MODEL:
TITLE 新产品的市场预测;
SETS:
  PROD/ A B C D/: P;
  LINK(PROD, PROD): T;
ENDSETS
DATA: ! 转移概率矩阵;
  T = .75  .1  .05  .1
      .4   .2  .1   .3
      .1   .2  .4   .3
      .2   .2  .3   .3;
ENDDATA
@FOR(PROD(I): P(I)=@SUM(LINK(J,I): P(J)* T(J,I)) );
```

```
@SUM(PROD: P) = 1;
@FOR(PROD(I): @WARN( '输入矩阵的每行之和必须是1', @ABS( 1 -
@SUM(LINK(I,J): T(I,J)))#GT# .000001));
END
```

可以指出的是，上面LINGO模型中最后的语句@WARN只是为了验证输入矩阵的每行之和必须是1，而且我们看到为了比较两个实数（如X和1）是否相等，一般不能直接用“X#NE#1”，因为受计算机字长（精度）的限制，实数在计算机内存存储是有误差的。所以，通常的方法是比较这两个实数之差的绝对值是否足够小。

求解结果为 A,B,C,D 的市场份额分别是47.5%，15.25%，16.75%，20.5%。

3.2　产品属性的效用函数

一般来讲，每种产品（如某种品牌的小汽车）都有不同方面的属性，例如价格、安全性、外观、保质期等。在设计和销售新产品之前，了解顾客对每种属性的各个选项的偏好程度非常重要。偏好程度可以用效用函数来表示，即某种属性的不同选项对顾客的价值（效用）。不幸的是，让顾客直接精确地给出每个属性的效用函数一般是困难的，例如对于价格，一般的顾客当然会说越便宜越好，但很难确定10万元的价格和15万元的价格的效用具体是多少。但是，对于具体的产品，产品的各个属性的具体选项配置都已经确定下来了，所以如果我们把一些具体的产品让顾客进行评估打分，顾客通常能比较容易地给出具体产品的效用。那么，从这些具体产品的效用信息中，我们能否反过来估计每个属性中各个选项的效用呢？这种方法通常称为联合分析（conjoint analysis）。下面通过一个例子来说明。

例11　对某种牌号的小汽车，假设只考虑两种属性：价格和安全气囊。价格分为12.9万元、9.9万元、7.9万元；安全气囊的配置为两个、一个、没有。经过市场调查，顾客对该产品的不同配置的偏好程度（效用）如表9所示（表中的值（权重）越大表示顾客越喜欢）。那么，价格和安全气囊的效用函数如何？

表9　顾客对产品的不同配置的偏好程度

价格（万元）	安　全　气　囊		
	2	1	0
12.9	7	3	1
9.9	8	4	2
7.9	9	6	5

（1）模型建立

记价格选项分别为H（高），M（中），L（低），对应的效用为 p_j（$j=$H，M，L）；安全气囊选项分别为0，1，2，对应的效用为 q_i（$i=0,1,2$）。我们的目的实际上就是

要求出 p_j 和 q_i。

假设价格和安全气囊的效用是线性可加的，即当价格选项为 j、安全气囊选项为 i 时，具体产品的效用 $c(i,j)$ 应该可以用价格的和安全气囊的效用之和来估计

$$c(i,j) = p_j + q_i \tag{40}$$

那么，如何比较不同的估计的好坏呢？一种简单的想法是针对6个待定参数（p_j 和 q_i），表中给出了9组数据，因此可以用最小二乘法确定 p_j 和 q_i。也就是说，此时的目标为

$$\min \sum_i \sum_j [c(i,j) - c_0(i,j)]^2 \tag{41}$$

其中，$c_0(i,j)$ 是表中的数据（安全气囊选项为 i、价格选项为 j 时具体产品的效用）。

因为做效用分析的主要目的是将来用于把不同配置的具体产品的优劣次序排出来，所以另一种方法是希望 $c(i,j)$ 和 $c_0(i,j)$ 保持同样的顺序：即对任意的 (i,j) 和 (k,l)，当 $c_0(i,j)+1 \le c_0(k,l)$ 时，也尽量有 $c(i,j)+1 \le c(k,l)$（这里"+1"表示 $c(i,j)$ 严格小于 $c(k,l)$，且至少相差1）。于是，可以考虑如下目标函数

$$\min \sum_{i,j} \sum_{k,l} \max\{1 + p_j + q_i - p_l - q_k, 0\} \tag{42}$$

式中的求和只是对满足 $c_0(i,j)+1 \le c_0(k,l)$ 的 (i,j) 和 (k,l) 求和。在LINGO中由于所有的变量默认的都为非负变量，程序中式（42）可以改写为

$$\min \sum_{i,j} \sum_{k,l} (1 + p_j + q_i - p_l - q_k)$$

（2）模型求解

LINGO程序如下：

```
MODEL:
TITLE 产品属性的效用函数;
SETS:
    PRICE/H,M,L/:P;
    SAFETY/2,1,0/:Q;
    M(safety,PRICE):CI;
    MM(M,M)|CI(&1,&2)+1 #LE# CI(&3,&4):ERROR;
```

```
ENDSETS
DATA:
    CI=7  8  9  3  4  6  1  2  5;
ENDDATA
@FOR(MM(i,j,k,l):ERROR(i,j,k,l)>1+P(j)+Q(i)-(P(l)+Q(k)));
[obj] MIN = @SUM(mm: ERROR);
END
```

求解这个模型，得到

$$p_H = 0，\ p_M = 1，\ p_L = 4，\ q_0 = 0，\ q_1 = 2，\ q_2 = 7$$

此时模型的最优值（误差和）为0，所以说明在这个效用函数下，虽然得到的产品权重（效用）与问题中给出的数据不完全相同，但产品的相对偏好顺序是完全一致的。

（3）模型的讨论

下面我们看看用最小二乘法确定 p_j 和 q_i 的结果是否与此相同。此时的模型实际上就是一个简单的二次规划模型。LINGO程序为

```
MODEL:
TITLE 最小二乘法计算产品属性的效用函数;
SETS:
    PRICE/H,M,L/:P;
    SAFETY/2,1,0/:Q;
    M(safety,PRICE):CI,ERROR,sort;
ENDSETS
DATA:
    CI =7  8  9  3  4  6  1  2  5;
ENDDATA
@FOR(M(i,j):sort(i,j)=p(j)+q(i);ERROR(i,j)= sort(i,j)-CI(i,j) );
MIN=@SUM(M:@sqr(ERROR));
@FOR(M(i,j): @FREE(ERROR) );
!@FOR(price:@gin(P));
!@FOR(safety:@gin(Q));
END
```

上面模型中的sort变量表示的就是按照这里新计算的效用函数得到的不同配置下的产品的效用。

通过运行LINGO程序，可以看到，此时的效用函数的结果与前面得到的结果不同，但仔细察看SORT的结果可以发现，不同配置产品之间的相对顺序仍然是保持的。

不过，最小二乘法得到的产品的效用是一些带有小数的数，实际中使用不太方便。如果希望得到整数解，只需要在模型中“END”语句前增加下面两行语句：

```
    @FOR(price:@gin(P));
```

```
@FOR(safety:@gin(Q));
```

求解结果中，SORT（1，M）＝SORT（0，L）＝4，这两个配置没能分辨出来。

综合这些讨论，结论还是我们在基本模型中给出的结果比较令人满意。请读者思考一下：基本模型中并没有要求决策变量取整数，为什么正好是整数？这是偶然的，还是必然的？

3.3 机票的销售策略

例12 某航空公司每天有三个航班服务于 A,B,C,H 四个城市，其中城市 H 是可供转机使用的。三个航班的出发地－目的地分别为 AH,HB,HC，可搭乘旅客的最大数量分别为120人，100人，110人，机票的价格分头等舱和经济舱两类。经过市场调查，公司销售部得到了每天旅客的相关信息，见表10。该公司应该在每条航线上分别分配多少头等舱和经济舱的机票？

表10 市场调查数据

出发地－目的地	头等舱需求（人）	头等舱价格（元）	经济舱需求（人）	经济舱价格（元）
AH	33	190	56	90
AB(经H转机)	24	244	43	193
AC(经H转机)	12	261	67	199
HB	44	140	69	80
HC	16	186	17	103

（1）问题分析

公司的目标应该是使销售收入最大化，由于头等舱的机票价格大于对应的经济舱的机票价格，很容易让人想到先满足所有头等舱的顾客需求：这样 AH 上的头等舱数量=33+24+12=69，HB 上的头等舱数量=24+44=68，HC 上的头等舱数量=12+16=28，等等，但这种贪婪算法是否一定得到最好的销售计划？

（2）模型建立

考虑5个起终点航线 AH, AB, AC, HB, HC，依次编号为 i（$i=1,2,\cdots,5$），相应的头等舱需求记为 a_i，价格记为 p_i；相应的经济舱需求记为 b_i，价格记为 q_i。此外，三个航班 AH, HB, HC 的顾客容量分别是 $c_1=120$，$c_2=100$，$c_3=110$。这就是例中给出的全部数据。

设航线i（$i=1,2,\cdots,5$）上销售的头等舱机票数为x_i，销售的经济舱机票数为y_i，这就是决策变量。

显然，目标函数应该是

$$\sum_{i=1}^{5}(p_i x_i + q_i y_i) \tag{43}$$

约束条件有以下两类：

i）三个航班上的容量限制

例如，航班AH上的乘客应当是购买AH, AB, AC机票的所有旅客，所以

$$\sum_{i=1}^{3}(x_i + y_i) \le c_1 \tag{44}$$

同理，有

$$x_2 + y_2 + x_4 + y_4 \le c_2 \tag{45}$$

$$x_3 + y_3 + x_5 + y_5 \le c_3 \tag{46}$$

ii）每条航线上的需求限制

$$0 \le x_i \le a_i,\ 0 \le y_i \le b_i,\ i = 1,2,\cdots,5 \tag{47}$$

（3）模型求解

```
MODEL:
TITLE 机票销售计划;
SETS:
    route /AH,AB,AC,HB,HC/:a,b,p,q,x,y;
ENDSETS
DATA:
a p b q=
```

33	190	56	90
24	244	43	193
12	261	67	199
44	140	69	80
16	186	17	103

```
;
c1 c2 c3 = 120 100 110;
ENDDATA
[obj] Max = @SUM(route: p*x+q*y );
```

```
[AH] @SUM(route(i)|i#ne#4#and#i#ne#5:x(i)+y(i)) < c1;
[HB] @SUM(route(i)|i#eq#2#or#i#eq#4:x(i)+y(i))  < c2;
[HC] @SUM(route(i)|i#eq#3#or#i#eq#5:x(i)+y(i))  < c3;
@FOR(route: @bnd(0,x,a);@bnd(0,y,b) );
END
```

计算结果为，航线 AH, AB, AC, HB, HC 上分别销售33，10，12，44，16张头等舱机票，分别销售0，0，65，46，17张经济舱机票，总销售收入为39344元。从三个约束的松弛/剩余（slack or surplus）均为0可知，机上已经全部满员。

（4）结果讨论

按道理，机票张数还应该有整数约束。这里直接按连续线性规划解，得到的解已经都是整数，所以也就没有必要再加上整数约束了。

最后我们指出：最优解中 AB 线路上头等舱的需求（24人）并没有全部得到满足，所以本节开始时介绍的贪婪算法的思想是不能保证求到最优解的。事实上，读者不难求出贪婪算法得到的解对应的总销售额是38854元，小于这里的最优值39344元。

习题二十六

1．某银行经理计划用一笔资金进行有价证券的投资，可供购进的证券及其信用等级、到期年限、收益如表11所示。按照规定，市政证券的收益可以免税，其它证券的收益需按50%的税率纳税。此外还有以下限制：

i）政府及代办机构的证券总共至少要购进400万元；

ii）所购证券的平均信用等级不超过1.4（信用等级数字越小，信用程度越高）；

iii）所购证券的平均到期年限不超过5年。

表11　证券相关的信息

证券名称	证券种类	信用等级	到期年限	到期税前收益（%）
A	市政	2	9	4.3
B	代办机构	2	15	5.4
C	政府	1	4	5.0
D	政府	1	3	4.4
E	市政	5	2	4.5

（1）若该经理有1000万元资金，应如何投资？

（2）如果能够以2.75%的利率借到不超过100万元资金，该经理应如何操作？

（3）在1000万元资金情况下，若证券 A 的税前收益增加为4.5%，投资应否改变？若证券 C 的税前收益减少为4.8%，投资应否改变？

2．假设某公司在银行有一个现金帐户和一个长期投资帐户，现金帐户利息很低，而长期投资帐户利息较高。所有业务往来（收入和支出）只能通过现金帐户进行，如果现金帐户中钱很多，就可能需要将一部分钱转入长期投资帐户；反之，需要将一部分钱

从长期投资帐户转入现金帐户。为简单起见，假设以万元为单位，现金帐户的钱数只能是－20，－10，0，…，40，50（万元）之一，分别记为状态1，2，…，7，8，它们每个月分别导致的费用如表12所示。此外，根据统计，如果当月现金帐户的状态位于i（$2\le i\le 7$），下个月现金帐户的状态只可能位于$i-1,i,i+1$三者之一，并且概率分别为0.4，0.1，0.5；如果当月现金帐户的状态位于1，则下个月现金帐户的状态只可能位于1和2，并且概率分别为0.5，0.5；如果当月现金帐户的状态位于8，则下个月现金帐户的状态只可能位于7和8，并且概率分别为0.4，0.6。

表12　现金帐户的钱数对应的成本

现金数量(万元)	-20	-10	0	10	20	30	40	50
状态	1	2	3	4	5	6	7	8
费用(万元)	1.4	0.7	0	0.2	0.4	0.6	0.8	1

每月初你可以改变当前状态（即从长期投资帐户转入现金帐户，或从现金帐户转入长期投资帐户），但假设每次状态的改变银行收取0.3万元的固定费用，此外还要收取转帐金额5％的转帐手续费。请你建立优化模型，确定如果当月现金帐户的状态位于i，是否应该改变当前状态，如何改变状态？